AD-A156 755   ADAPTIVE FILTERING FOR SEISMIC SIGNAL DETECTION PART 1   1/1
(U) SOUTHEASTERN MASSACHUSETTS UNIV NORTH DARTMOUTH
DEPT OF ELECT.. C H CHEN 11 OCT 79 SMU-EE-TR-5-PT-1
UNCLASSIFIED   DACW39-79-M-2766   F/G 17/10   NL

END
FILMED
DTIC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

AD-A156 755

Final Report (Part I)
Contract Number DACW 39-79-M-2766
SMU-EE-TR-5
October 11, 1979

ADAPTIVE FILTERING FOR

SEISMIC SIGNAL DETECTION

Prepared by

Dr. C. H. Chen
Department of Electrical Engineering
Southeastern Massachusetts University
North Dartmouth, Massachusetts 02747

Submitted to

Dr. Daniel Cress
Environmental Constraints Group
Waterways Experiment Station,
Corps of Engineers
Department of the Army
P. O. Box 631
Vicksburg, Mississippi 39180

ADAPTIVE FILTERING FOR SEISMIC SIGNAL DETECTION

Final Report (Part I)

I. Statement of the Problem

This project is concerned with the adaptive digital processing to remove (or reduce) noisy signals generated by sensors at large distance away from the sensors. The noisy signals need to be removed because only those signsals generated near the sensors are of interest for detection. The signals to be detected may correspond to footsteps of a walking man. The practical approach to this problem at the Waterways Experiment Station is to use two or more spatially separated sensors. Each sensor may consist of a number of transducer elements-each contributing constructively to the total response of the sensor. Figure A is a sketch of the sensor layout. In this sketch, the seismic energy from the far-field source is very similar at both sensors 1 and 2. Therefore, we should be able to reduce the final signal used for processing by removing the correlated portions of the signal observed at sensors 1 and 2. When a person is very near to sensor 1 (sensor 2 being 3 meters away or more), he generates a much different signal in sensor 1 than in sensor 2. Therefore, the signals of interest should not be correlated. Mathematically the problem is to detect the footstep signal from the correlated background noise caused by e.g. moving vehicles at some distance away plus additive sensor noises. The adaptive procedure is needed because of the non-stationary nature of the data considered.

II. Methods of Approach

Two classes of digital processing techniques are considered; namely the adaptive Kalman filtering and the adaptive digital filtering. The adaptive Kalman filter continuously monitors the "state jumps" due to the presence of personnel. Without state jumps the filter performs as an ordinary Kalman filter. The filter parameters are adjusted if a state jump is detected. False detection can be minimized by averaging over a sufficiently large "window." A generalized likelihood ratio is used for detection of state jumps. The method was originally proposed by Willsky and Jones [1] and was successfully employed by us in detecting the object boundary in image scan lines [2].

The block diagram of the adaptive digital filter considered is shown in Figure B. The input-output relation of the non-recursive Adaptive Digital Filter (ADP) is given by (Figure C),

$$g_m = \sum_{n=0}^{N} b_{n,m} f_{m-n} \tag{1}$$

where g_m is the filter output, $b_{n,m}$ is the filter coefficient, and f_n is the filter input. Define the vectors

$$B_m' = [b_{o,m} \;\; b_{1,m} \;\; \cdots \;\; b_{N,m}]$$

and

$$F_m' = [f_m \;\; f_{m-1} \;\; \cdots \;\; f_{m-N}]$$

where N is the order of the filter, and m is the number of iterations in adjusting the filter coefficients. Eq. (1) can be written as

$$g_m = B_m' F_m = F_m' B_m \tag{2}$$

We use the Noisy LMS Algorithm [3] in our computer program to calculate the filter coefficients $b_{n,m}$, i.e.

$$
\begin{cases}
B_{m+1} = B_m + v e_m F_m \\
e_m = y_m - g_m \\
g_m = B_m' F_m
\end{cases}
\tag{3}
$$

where v is a constant. After adjusting the values of v, N, m, and delay time, we can obtain the desired result with a greatly improved signal-to-noise ratio.

III. Implementation and Computer Results

Extensive effort was made to implement the two methods described above on the PDP 11/45 computer with the seismic data supplied by the Waterways Experiment Station. The data are stored in two magnetic tapes. The first tape consists of 4-channel data while the second tape consists of 14-channel data. Mr. Mark Flohr of the Waterways Experiment Station was particularly helpful to acquaint us with the structure of the data format during his visit in August 1979. The computer results presented are based on a data set of footstep signal plus a small amount of noise. To cope with the memory capacity of the minicomputer, every fifth data sample is taken from the 4-channel data tape with a resulting sampling rate of 200 samples/sec. A sequence of 960 sample points are taken for detailed study (Record 327 to 356). Figures 1a, 2a, 3a and 4a show the digitized data from the four channels with corresponding Fourier amplitude spectra given by Figures 1b, 2b, 3b and 4b, respectively.

The Adaptive Kalman Filtering was implemented to detect the jumps. Figures 5, 6, 7 and 8 show the waveforms before and after detection along with a printout of the number of jumps detected. These numbers are generally correct. The complete listing of the computer program has been forwarded to the Waterways Experiment Station. Further improvement of the program is desirable to demonstrate the effect of noise reduction while performing the jump detection. The parameters are fairly sensitive to the result and the following list is the best set of parameters:

AA (transition coefficient) = 1.0

Q (covariance of the system driving noise) = 1.0

R (noise power) = 1.0

TH (threshold) = 10.0

M (length of window) = 10

There is no estimation of parameter R or transition matrices W and P. Total number of samples is 960 per channel.

For the implementation of the Adaptive Digital Filter on the PDP 11/45, the subroutine AF serves as the Adaptive Filter, and the subroutine SHIFTR serves as the Shift-Right Register. The meanings of all variables and arrays in the subroutine AF are stated below.

ND:	Noisy data array, i.e. input data
NF:	Number of noisy data points
NDELAY:	Number of delay points
M:	Number of iterations
N:	The order of the ADF
V:	The constant v of the ADF
B:	Coefficients of the ADF, stored in an array
DND:	Delay elements for storing delayed data points
F:	The variables f_m in the ADF, as an array
G:	The variables g_m in the ADF, i.e., the output data
E:	The variables e_m in the ADF
Y:	The variable y

Array B and array F are initially set to zero. The program processes the data on a channel-by-channel basis. The complete program named as SPAN4 also includes an ANFFT Subroutine which performs FFT on the original input data and also on the filtered output data. The complete program listing is in the Appendix.

Figure 9 shows a sequence of filtered waveforms for channel 1. The best result is given by Figure 9e with a corresponding FFT given by Figure 9f. Here M = 1, NDELAY = 30, N = 18, and V = 0.0112. Figure 10 shows a sequence of filtered waveforms for channel 2. The best result is given by Figure 10e with a corresponding FFT given by Figure 10f. Here M = 1, NDELAY = 30, N = 17 and V = 0.00145. It is noted that the parameter choice is also very sensitive in the adaptive digital filtering operations. The filtered results clearly illustrate the suppression of background noise and the enhancement of footstep signals. The filtered results are so good that further process of jump detection appears unnecessary. All Fourier spectra correspond to a frequency range of 0 to 100 Hz.

Both adaptive filtering techniques have been shown to be very effective for the seismic detection tasks under consideration. They both are computationally efficient. Proper choice of parameters is an important step to derive the best results. Although mathematically the Kalman filter and the adaptive digital filter are somewhat equivalent, the implementations of the two techniques are different and thus the results are different also. The adaptive Kalman filtering in fact has more flexibility to adjust itself to the change in data characteristics.

References

1. A. J. Willsky and H. L. Jones, "A Generalized Likelihood Ratio Approach to the Detection and Estimation of Jumps in Linear Systems," IEEE Trans. on Automatic Control, February, 1976.

2. C. H. Chen, "Adaptive Image Filtering," Proc. of 1979 IEEE Pattern Recognition and Image Processing Conference, August, 1979.

3. N. Ahmed, "A Study of Adaptive Digital Filters," Report SAND 77-0102, August, 1977.

Acknowledgement

Among all graduate research assistants participating in the project, Mr. Jenshiun Chen has made a major contribution to the implementation of both adaptive Kalman filtering and adaptive digital filtering. Mr. Chihsung Yen has also been very helpful in the software development.

Appendix

Program SPAN4

```
      DIMENSION F(4,32),X(1024),Y(1024),C(257)
      COMPLEX CY(1024)
      INTEGER CHNO,PREF,DLY,ITR,ODR
      COMMON F,NF,INDEX
      DEFINE FILE 3(30,256,U,INDEX)
6     FORMAT('0')
5     FORMAT(' INPUT IGD,NP,PREF,DLY,ITR,ODR,NTE,V')
7     FORMAT(7I5,F12.5)
      WRITE(6,5)
      WRITE(6,6)
      READ(6,7)IGD,NP,PREF,DLY,ITR,ODR,NTE,V
      IF(IGD.NE.0)GO TO 71
      CALL GDATA

71    NF=960
      DO 8 I=1,1024
8     X(I)=FLOAT(I)
      DO 10 CHNO=1,4
      INDEX=1
      DO 112 J=1,30
      READ(3'INDEX)F
      L=(J-1)*32
      DO 15 I=1,32
      L=L+1
      Y(L)=F(CHNO,I)
15    CONTINUE
112   CONTINUE
      DO 352 I=1,NF
      Y(I)=Y(I)/1000
352   CONTINUE
      DO 419 I=961,1024
      Y(I)=0
419   CONTINUE
20    FORMAT(' ***** PLOT OF ORIGINAL DATA CHANNEL',I3)
      IF(NP.NE.0)GO TO 56
      WRITE(5,54)CHNO
54    FORMAT('1******** ORIGINAL DATA FUNCTION VALUES OF CHANNEL ',I2)
      WRITE(5,55)Y
55    FORMAT(1X,10F13.5)
56    WRITE(6,21)NF
21    FORMAT(' NFT=',I4)
      READ(6,30)NX
      CALL NEWPAG
      WRITE(6,20)CHNO
      WRITE(6,21)NF
30    FORMAT(I3)
      CALL FRPLOT(X,Y,1024,0,1023,10,700)
      READ(6,30)NX
      CALL NEWPAG
      IF(PREF.NE.1)GO TO 931
      CALL AF(Y,NF,DLY,ITR,ODR,NTE,V)

931   DO 939 I=1,1024
      CY(I)=CMPLX(Y(I),0.0)
939   CONTINUE
```

```
      IF(PREF.EQ.0)GO TO 2100
      WRITE(6,87)CHNO
87    FORMAT('1******* PLOT OF  PREFILTERED DATA OF CHANNEL',I2)
      CALL KBPLOT(X,Y,1024,1,1000,10,700)
      READ(6,30)NX
      CALL NEWPAG
      IF(NP.NE.0)GO TO 2100
      WRITE(5,2110)CHNO
2110  FORMAT('1******* PREFILTERED DATA OF CHANNEL',I2)
      WRITE(5,55)Y
2100  CALL ANFFT(CY,10)
      DO 950 I=1,1024
      Y(I)=CABS(CY(I))
950   CONTINUE
      WRITE(6,50)CHNO
50    FORMAT(' ***** TYPE RERUTN FOR PLOT OF FFT DATA  CHANNEL',I2)
      READ(6,30)NX
      CALL KBPLOT(X,Y,512,0,1023,10,700)
      READ(6,30)NX
      IF(NP.NE.0)GO TO 175
      WRITE(5,171)CHNO
171   FORMAT('1******** FFT DATA FUNCTION  VALUES OF CHANNEL ',I2)
      WRITE(5,55)Y
175   READ(6,30)NX
10    CONTINUE
      CALL EXIT
      END
      SUBROUTINE GDATA
      INTEGER CH,RATE
      BYTE B(1280)
      DIMENSION IA(1280),F(4,32)
      COMMON F,NF,INDEX
      INDEX=1
      DO 200 NR=1,30
      CALL READUN(B,NPT)
      NF=NPT/8*6
      WRITE(6,50)NPT
50    FORMAT('OTOTAL BYTES READ FROM TAPE =',I7)
      DO 70 I=1,NPT
      CALL BY2IN(B(I),II)
      IA(I)=II
70    CONTINUE
85    FORMAT(12I5)
      DO 80 I=1,NPT,2
      IA((I+1)/2)=IA(I)*256+IA(I+1)
80    CONTINUE
      NT=NPT/2
      DO 90 CH=1,4
      K=1
      DO 90 I=CH,NT,20
      F(CH,K)=FLOAT(IA(I))
90    K=K+1
      WRITE(3'INDEX)F
200   CONTINUE
      RETURN
      END
```

```
      SUBROUTINE ANFFT(X,M)
      COMPLEX X(1),U,W,T
      N=2**M
      N2=N/2
      N1=N-1
      J=1
      DO 3 I=1,N1
      IF(I.GE.J)GO TO 1
      T=X(J)
      X(J)=X(I)
      X(I)=T
1     K=N2
2     IF(K.GE.J)GO TO 3
      J=J-K
      K=K/2
      GO TO 2
3     J=J+K
      PI=3.1415926
      DO 5 L=1,M
      LE=2**L
      LE1=LE/2
      U=(1.0,0.0)
      W=CMPLX(COS(PI/LE1),SIN(PI/LE1))
      DO 5 J=1,LE1
      DO 4 I=J,N,LE
      ID=I+LE1
      T=X(ID)*U
      X(ID)=X(I)-T
4     X(I)=X(I)+T
5     U=U*W
      RETURN
      END
      SUBROUTINE AF(ND,NF,NDELAY,M,N,NTE,V)
      REAL ND(1),F(40),B(40),DND(400)
      DOUBLE PRECISION G
      G=0
      NP=N+1
      DO 50 I=1,NP
      B(I)=0.
      F(I)=0.
50    CONTINUE
      DO 40 I=1,NDELAY
      J=NDELAY-I+1
      DND(J)=ND(I)
40    CONTINUE
```

```
      IIP=NDELAY+1
      DO 100 IND=IIP,NF
      Y=ND(IND)
      CALL SHIFTR(Y,DND,NDELAY,TEMP)
      CALL SHIFTR(TEMP,F,NP,TEMP1)
      DO 200 IM=1,M
      G=0
      DO 500 I=1,NP
      G=G+B(I)*F(I)
500   CONTINUE
      E=Y-G
      DO 400 I=1,NP
      B(I)=B(I)+V*E*F(I)
400   CONTINUE
      IF(NTE.NE.0)GO TO 200
      PRINT 600,IND,IM,E,G
600   FORMAT(1X,'SAMPLE',I5,'ITERATION',I5,'E=',F15.5,'G=',F15.5)
      PRINT 601,B
601   FORMAT(1X,12F10.2)
      PRINT 602,F
602   FORMAT(1X,12F10.2)
200   CONTINUE
700   ND(IND)=G
100   CONTINUE
      RETURN
      END
      SUBROUTINE SHIFTR(SHIN,A,L,SOUT)
      REAL A(1)
      SOUT=A(L)
      NDIM=L-1
      DO 100 K=1,NDIM
      I=L-K+1
      J=I-1
      A(I)=A(J)
100   CONTINUE
      A(1)=SHIN
      RETURN
      END
```



Figure A



Figure B



Figure C

Rec 327-356



Fig. 1a



Fig. 1b



Fig. 2a



Fig. 2b

Fig.3a



Fig 3b



Fig.4a



Fig 4b



Fig 5





Fig. 5

Fig. 6





Fig 6

Fig. 7





Fig 7

Fig. 8





Fig. 8

Fig. 9a



DLY = 30 ZTR = 1 , ODR = 16 , V = .0119



Fig. 9b

DLY ~ 30 ZTR = 1 , ODR = 17 , V = .0119



Fig. 9c

DLY = 30 · ZTR = 1 ODR = 17 V = .0119



Fig. 9d

DLY = 30 ITR=1 , ODR=18 V=.0.12



Fig 9a

DLY=30 ITR=1 ODR=18 V= .012



Fig 9f



Fig 10a

DLY = 30 ITR=1 , ODR=17 V=.002



Fig 10b

DLY=30 ZTR=1 ODR=17 V=.0015



Fig.10c

DLY=30 ZTR=1 ODR=17 V=.0014



Fig 10d

DLY=30 ZTR=1 ODR=17 V=.00145



Fig.10e

DLY=30 ZTR=1 ODR=17 V=.00145



Fig 10f

END

FILMED

8-85

DTIC